

1 Part A

1.1 sum.y Iteratively sum linked list elements

Solution The struct ELE used in the sum_list() function shown on listing 1.

```
1  /* $begin examples */
2  /* linked list element */
3  typedef struct ELE {
4      long val;
5      struct ELE *next;
6  } *list_ptr;
```

Listing 1: ELE.c

The function sum_list() written in C displayed on listing 2.

```
1  /* sum_list - Sum the elements of a linked list */
2  long sum_list(list_ptr ls)
3  {
4      long val = 0;
5      while (ls) {
6          val += ls->val;
7          ls = ls->next;
8      }
9      return val;
10 }
```

Listing 2: sum_list.c

Full example of sum.y with stack initialization, test data, and main function depicted on listing 3.

```
1  # Execution begins at address 0
2      .pos 0
3      irmovq stack, %rsp          # Set up stack pointer
4      call main                   # Execute main program
5      halt                       # Terminate program
6
7  # Sample linked list
8      .align 8
9  ele1:
10     .quad 0x00a
11     .quad ele2
12  ele2:
13     .quad 0x0b0
14     .quad ele3
15  ele3:
16     .quad 0xc00
17     .quad 0
18
19  main:
20     irmovq ele1,%rdi
21     call sum_list                # sum_list(ele1)
22     ret
23
24  # long sum_list(list *ls)
25  # ls in %rdi
26  sum_list:
27     irmovq $0,%rax
28     andq %rdi,%rdi
29     je test
30     loop:
31     mrmovq (%rdi), %rsi
32     addq %rsi,%rax
33     mrmovq 8(%rdi),%rdi
34     andq %rdi,%rdi
35     jne loop
36     test: ret
37
38  # Stack starts here and grows to lower addresses
39     .pos 0x200
40  stack:
```

Listing 3: sum.y

1.2 rsum.y_s Recursively sum linked list elements

Solution The same ELE struct from listing 1 used in the `rsum_list()` function. The C version of `rsum_list()` is on listing 4.

```
1  /* rsum_list - Recursive version of sum_list */
2  long rsum_list(list_ptr ls)
3  {
4      if (!ls)
5          return 0;
6      else {
7          long val = ls->val;
8          long rest = rsum_list(ls->next);
9          return val + rest;
10     }
11 }
```

Listing 4: `rsum_list.c`

Full version of `rsum.ys` is on listing 5.

```
1  # Execution begins at address 0
2      .pos 0
3      irmovq stack, %rsp
4      call main
5      halt
6
7  # Sample linked list
8      .align 8
9  ele1:
10     .quad 0x00a
11     .quad ele2
12  ele2:
13     .quad 0x0b0
14     .quad ele3
15  ele3:
16     .quad 0xc00
17     .quad 0
18
19  main:
20     irmovq ele1,%rdi
21     call rsum_list
22     ret
23
24  # long rsum_list(list *ls)
25  # ls in %rdi
26  rsum_list:
27     pushq %r12
28     irmovq $0,%rax
29     andq %rdi,%rdi
30     je test
31     mrmovq (%rdi), %r12
32     mrmovq 8(%rdi),%rdi
33     call rsum_list
34     addq %r12,%rax
35     test: popq %r12
36     ret
37
38  # Stack starts here and grows to lower addresses
39     .pos 0x200
40  stack:
```

Listing 5: `rsum.ys`

1.3 copy.yy Copy a source block to a destination block

Solution The C version of `copy_block()` is on listing 6.

```
1  /* copy_block - Copy src to dest and return xor checksum of src */
2  long copy_block(long *src, long *dest, long len)
3  {
4      long result = 0;
5      while (len > 0) {
6          long val = *src++;
7          *dest++ = val;
8          result ^= val;
9          len--;
10     }
11     return result;
12 }
```

Listing 6: `copy_block.c`

Final version of `copy.yy` is on listing 7.

```
1  # Execution begins at address 0
2      .pos 0
3      irmovq stack, %rsp          # Set up stack pointer
4      call main                   # Execute main program
5      halt                       # Terminate program
6
7      .align 8
8  # Source block
9  src:
10     .quad 0x00a
11     .quad 0x0b0
12     .quad 0xc00
13 # Destination block
14 dest:
15     .quad 0x111
16     .quad 0x222
17     .quad 0x33
18
19 main:
20     irmovq $3,%rdx
21     irmovq dest,%rsi
22     irmovq src,%rdi
23     call copy_block              # copy_block(src, dest, 3)
24     ret
25
26 # long copy_block(long *src, long *dest, long len)
27 copy_block:
28     irmovq $0,%rax
29     andq %rdx,%rdx
30     jle test
31     loop: mrmovq (%rdi),%r10
32     irmovq $8,%r11
33     addq %r11,%rdi
34     rmmovq %r10,(%rsi)
35     addq %r11,%rsi
36     xorq %r10,%rax
37     irmovq $1,%r11
38     subq %r11,%rdx
39     andq %rdx,%rdx
40     jg loop
41     test: ret
42
43 # Stack starts here and grows to lower addresses
44     .pos 0x200
45 stack:
```

Listing 7: `copy.yy`

2 Part B

Solution The updated hcl description of SEQ control signals for implementing IADDQ instruction show on listing 8.

```

1 ##### Fetch Stage #####
2
3 # Determine instruction code
4 word icode = [
5     imem_error: INOP;
6     1: imem_icode; # Default: get from instruction memory
7 ];
8
9 # Determine instruction function
10 word ifun = [
11     imem_error: FNONE;
12     1: imem_ifun; # Default: get from instruction memory
13 ];
14
15 bool instr_valid = icode in
16     { INOP, IHALT, IRRMOVQ, IIRMOVQ, IRMMOVQ, IMRMVQ,
17       IOPQ, IJXX, ICALL, IRET, IPUSHQ, IPOPQ, IIADDQ }; ##### Add IIADDQ
18       icode to instr_valid
19
20 # Does fetched instruction require a regid byte?
21 bool need_regids =
22     icode in { IRRMOVQ, IOPQ, IPUSHQ, IPOPQ,
23               IIRMOVQ, IRMMOVQ, IMRMVQ, IIADDQ }; ##### Add IIADDQ icode to
24               need_regids
25
26 # Does fetched instruction require a constant word?
27 bool need_valC =
28     icode in { IIRMOVQ, IRMMOVQ, IMRMVQ, IJXX, ICALL, IIADDQ }; ##### Add IIADDQ
29     icode to need_valC
30
31 ##### Decode Stage #####
32
33 ## What register should be used as the A source?
34 word srcA = [
35     icode in { IRRMOVQ, IRMMOVQ, IOPQ, IPUSHQ } : rA;
36     icode in { IPOPQ, IRET } : RRSP;
37     1 : RNONE; # Don't need register
38 ];
39
40 ## What register should be used as the B source?
41 word srcB = [
42     icode in { IOPQ, IRMMOVQ, IMRMVQ, IIADDQ } : rB; ##### Add IIADDQ icode
43     to srcB
44     icode in { IPUSHQ, IPOPQ, ICALL, IRET } : RRSP;
45     1 : RNONE; # Don't need register
46 ];
47
48 ## What register should be used as the E destination?
49 word dstE = [
50     icode in { IRRMOVQ } && Cnd : rB;
51     icode in { IIRMOVQ, IOPQ, IIADDQ } : rB; ##### Add IIADDQ to dstE
52     icode in { IPUSHQ, IPOPQ, ICALL, IRET } : RRSP;
53     1 : RNONE; # Don't write any register
54 ];
55
56 ## What register should be used as the M destination?
57 word dstM = [
58     icode in { IMRMVQ, IPOPQ } : rA;
59     1 : RNONE; # Don't write any register
60 ];
61
62 ##### Execute Stage #####
63
64 ## Select input A to ALU
65 word aluA = [
66     icode in { IRRMOVQ, IOPQ } : valA;
67     icode in { IIRMOVQ, IRMMOVQ, IMRMVQ, IIADDQ } : valC; ##### Add IIADDQ to
68     aluA
69     icode in { ICALL, IPUSHQ } : -8;
70     icode in { IRET, IPOPQ } : 8;
71     # Other instructions don't need ALU

```

```

67 ];
68
69 ## Select input B to ALU
70 word aluB = [
71     icode in { IRMMOVQ, IMRMVQ, IOPQ, ICALL,
72             IPUSHQ, IRET, IPOPQ, IIADDQ } : valB;    ##### Add IIADDQ to aluB
73     icode in { IRRMOVQ, IIRMOVQ } : 0;
74     # Other instructions don't need ALU
75 ];
76
77 ## Set the ALU function
78 word alufun = [
79     icode == IOPQ : ifun;
80     1 : ALUADD;
81 ];
82
83 ## Should the condition codes be updated?
84 bool set_cc = icode in { IOPQ, IIADDQ };    ##### Add IIADDQ to set_cc
85
86 ##### Memory Stage #####
87
88 ## Set read control signal
89 bool mem_read = icode in { IMRMVQ, IPOPQ, IRET };
90
91 ## Set write control signal
92 bool mem_write = icode in { IRMMOVQ, IPUSHQ, ICALL };
93
94 ## Select memory address
95 word mem_addr = [
96     icode in { IRMMOVQ, IPUSHQ, ICALL, IMRMVQ } : valE;
97     icode in { IPOPQ, IRET } : valA;
98     # Other instructions don't need address
99 ];
100
101 ## Select memory input data
102 word mem_data = [
103     # Value from register
104     icode in { IRMMOVQ, IPUSHQ } : valA;
105     # Return PC
106     icode == ICALL : valP;
107     # Default: Don't write anything
108 ];
109
110 ## Determine instruction status
111 word Stat = [
112     imem_error || dmem_error : SADR;
113     !instr_valid : SINS;
114     icode == IHALT : SHLT;
115     1 : SAOK;
116 ];
117
118 ##### Program Counter Update #####
119
120 ## What address should instruction be fetched at
121
122 word new_pc = [
123     # Call. Use instruction constant
124     icode == ICALL : valC;
125     # Taken branch. Use instruction constant
126     icode == IJXX && Cnd : valC;
127     # Completion of RET instruction. Use value from stack
128     icode == IRET : valM;
129     # Default: Use incremented PC
130     1 : valP;
131 ];
132 /* $end seq-all-hcl */

```

Listing 8: Update SEQ control signals for IADDQ instruction

3 Part C

Solution The optimized version of function `ncopy()` is shown on 9. Average CPE after optimization is **7.98**.

```
1  /* $begin ncopy-ys */
2  #####
3  # ncopy.ys - Copy a src block of len words to dst.
4  # Return the number of positive words (>0) contained in src.
5  #
6  # Include your name and ID here.
7  #
8  # Describe how and why you modified the baseline code.
9  #
10 #####
11 # Do not modify this portion
12 # Function prologue.
13 # %rdi = src, %rsi = dst, %rdx = len
14 ncopy:
15
16 #####
17 # You can modify this portion
18 # Loop header
19     xorq %rax, %rax
20     iaddq $-5, %rdx
21     jl EndUn
22
23 LoopUn:
24     mrmovq (%rdi), %r10
25     mrmovq 8(%rdi), %r11
26     mrmovq 16(%rdi), %r12
27     mrmovq 24(%rdi), %r13
28     mrmovq 32(%rdi), %r9
29     rmmovq %r10, (%rsi)
30     rmmovq %r11, 8(%rsi)
31     rmmovq %r12, 16(%rsi)
32     rmmovq %r13, 24(%rsi)
33     rmmovq %r9, 32(%rsi)
34
35     andq %r10, %r10
36     jle First
37     iaddq $1, %rax
38 First:
39     andq %r11, %r11
40     jle Second
41     iaddq $1, %rax
42 Second:
43     andq %r12, %r12
44     jle Third
45     iaddq $1, %rax
46 Third:
47     andq %r13, %r13
48     jle Fourth
49     iaddq $1, %rax
50 Fourth:
51     andq %r9, %r9
52     jle Fifth
53     iaddq $1, %rax
54 Fifth:
55
56     iaddq $40, %rdi    # src++
57     iaddq $40, %rsi    # dst++
58     iaddq $-5, %rdx    # len--
59     jge LoopUn
60
61 EndUn:
62     iaddq $5, %rdx
63     je Done
64     iaddq $-1, %rdx
65     je One
66     iaddq $-1, %rdx
67     je Two
68     iaddq $-1, %rdx
69     je Three
70
71     mrmovq (%rdi), %r10
```

```

72     mrmovq 8(%rdi), %r11
73     mrmovq 16(%rdi), %r12
74     mrmovq 24(%rdi), %r13
75     rmmovq %r10, (%rsi)
76     rmmovq %r11, 8(%rsi)
77     rmmovq %r12, 16(%rsi)
78     rmmovq %r13, 24(%rsi)
79
80     andq %r10, %r10
81     jle F1
82     iaddq $1, %rax
83 F1:
84     andq %r11, %r11
85     jle F2
86     iaddq $1, %rax
87 F2:
88     andq %r12, %r12
89     jle F3
90     iaddq $1, %rax
91 F3:
92     andq %r13, %r13
93     jle Done
94     iaddq $1, %rax
95     jmp Done
96
97 Three:
98     mrmovq (%rdi), %r10
99     mrmovq 8(%rdi), %r11
100    mrmovq 16(%rdi), %r12
101    rmmovq %r10, (%rsi)
102    rmmovq %r11, 8(%rsi)
103    rmmovq %r12, 16(%rsi)
104
105    andq %r10, %r10
106    jle Th1
107    iaddq $1, %rax
108 Th1:
109    andq %r11, %r11
110    jle Th2
111    iaddq $1, %rax
112 Th2:
113    andq %r12, %r12
114    jle Done
115    iaddq $1, %rax
116    jmp Done
117 Two:
118    mrmovq (%rdi), %r10
119    mrmovq 8(%rdi), %r11
120    rmmovq %r10, (%rsi)
121    rmmovq %r11, 8(%rsi)
122    andq %r10, %r10
123    jle T1
124    iaddq $1, %rax
125 T1:
126    andq %r11, %r11
127    jle Done
128    iaddq $1, %rax
129    jmp Done
130 One:
131    mrmovq (%rdi), %r10
132    rmmovq %r10, (%rsi)
133    andq %r10, %r10
134    jle Done
135    iaddq $1, %rax
136 #####
137 # Do not modify the following section of code
138 # Function epilogue.
139 Done:
140     ret
141 #####
142 # Keep the following label at the end of your function
143 End:
144 #/* $end ncopy-ys */

```

Listing 9: Optimized version of ncopy.asm